Example shown for short time solution. See equation 20. Whole file not used.

```
> restart:with(inttrans):with(plots):
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2);
```

$$eq := \frac{\partial}{\partial t}\, u(x, t) = \frac{\partial^2}{\partial x^2}\, u(x, t) \tag{1}$$

```
> u(x,0):=0;
```

$$u(x, 0) := 0 \tag{2}$$

```
> bc1:=diff(u(x,t),x)=0;
```

$$bc1 := \frac{\partial}{\partial x}\, u(x, t) = 0 \tag{3}$$

```
> bc2:=u(x,t)=1;
```

$$bc2 := u(x, t) = 1 \tag{4}$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\, U(x) = \frac{d^2}{dx^2}\, U(x) \tag{5}$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(diff(laplace(u(x,t),t,s),x)=D(U)(0),laplace(u(x,t),t,
  s)=U(0),bc1);
```

$$bc1 := D(U)(0) = 0 \tag{6}$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(diff(laplace(u(x,t),t,s),x)=D(U)(1),laplace(u(x,t),t,
  s)=U(1),bc2);
```

$$bc2 := U(1) = \frac{1}{s} \tag{7}$$

```
> U(x):=rhs(dsolve({eqs,bc1,bc2},U(x)));
```

$$U(x) := \frac{e^{\sqrt{s}\, x}}{s\left(e^{\sqrt{s}} + e^{-\sqrt{s}}\right)} + \frac{e^{-\sqrt{s}\, x}}{s\left(e^{\sqrt{s}} + e^{-\sqrt{s}}\right)} \tag{8}$$

```
> U1s:=exp(s^(1/2))/s/(exp(s^(1/2))^2+1)*exp(s^(1/2)*x);
```

$$U1s := \frac{e^{\sqrt{s}}\, e^{\sqrt{s}\, x}}{s\left(\left(e^{\sqrt{s}}\right)^2 + 1\right)} \tag{9}$$

```
> U2s:=exp(s^(1/2))/s/(exp(s^(1/2))^2+1)*exp(-s^(1/2)*x);
```

$$U2s := \frac{e^{\sqrt{s}}\, e^{-\sqrt{s}\, x}}{s\left(\left(e^{\sqrt{s}}\right)^2 + 1\right)} \tag{10}$$

```
> U1S:=series(subs(exp(s^(1/2))=1/S,U1s),S):
> U1S:=subs(S=exp(-s^(1/2)),U1S):
> simplify(U1S);
```

$$\frac{e^{\sqrt{s}\,(x-1)} - e^{\sqrt{s}\,(x-3)} + e^{\sqrt{s}\,(x-5)} + O\left(e^{-7\sqrt{s}}\right) s}{s} \tag{11}$$

```
> U1S:=Sum((-1)^(n-1)*exp(s^(1/2)*(x-2*n+1))/s,n=1..infinity);
```

$$U1S := \sum_{n=1}^{\infty} \frac{(-1)^{n-1} e^{\sqrt{s}\,(x-2n+1)}}{s} \quad \textbf{(12)}$$

```
> u1s:=(-1)^(n-1)*exp(s^(1/2)*(x-2*n+1))/s;
```

$$u1s := \frac{(-1)^{n-1} e^{\sqrt{s}\,(x-2n+1)}}{s} \quad \textbf{(13)}$$

```
> u1t:=invlaplace(u1s,s,t);
```

$$u1t := (-1)^{1+n}\, invlaplace\left(\frac{e^{\sqrt{s}\,(x-2n+1)}}{s}, s, t\right) \quad \textbf{(14)}$$

```
> U1t:=Sum(u1t,n=1..infinity);
```

$$U1t := \sum_{n=1}^{\infty} (-1)^{1+n}\, invlaplace\left(\frac{e^{\sqrt{s}\,(x-2n+1)}}{s}, s, t\right) \quad \textbf{(15)}$$

```
> U2S:=series(subs(exp(s^(1/2))=1/S,U2s),S):
> U2S:=subs(S=exp(-s^(1/2)),U2S):
> simplify(U2S);
```

$$\frac{e^{-\sqrt{s}\,(x+1)} - e^{-\sqrt{s}\,(x+3)} + e^{-\sqrt{s}\,(x+5)} + O\left(e^{-7\sqrt{s}}\right) s}{s} \quad \textbf{(16)}$$

```
> U2S:=Sum((-1)^(n-1)*exp(-s^(1/2)*(x+2*n-1))/s,n=1..infinity);
```

$$U2S := \sum_{n=1}^{\infty} \frac{(-1)^{n-1} e^{-\sqrt{s}\,(x+2n-1)}}{s} \quad \textbf{(17)}$$

```
> u2s:=(-1)^(n-1)*exp(-s^(1/2)*(x+2*n-1))/s:
> u2t:=invlaplace(u2s,s,t);
```

$$u2t := (-1)^{1+n}\, invlaplace\left(\frac{e^{(-x-2n+1)\sqrt{s}}}{s}, s, t\right) \quad \textbf{(18)}$$

```
> U2t:=Sum(u2t,n=1..infinity);
```

$$U2t := \sum_{n=1}^{\infty} (-1)^{1+n}\, invlaplace\left(\frac{e^{(-x-2n+1)\sqrt{s}}}{s}, s, t\right) \quad \textbf{(19)}$$

The short time solution for the same problem can be otained using the methodology described in seciton 8.1.4 (examples 8.5 and 8.6) as:

```
> Ut:=U1t+U2t;
```

$$Ut := \sum_{n=1}^{\infty} (-1)^{1+n}\, invlaplace\left(\frac{e^{\sqrt{s}\,(x-2n+1)}}{s}, s, t\right) + \sum_{n=1}^{\infty} (-1)^{1+n}\, invlaplace\left(\frac{e^{(-x-2n+1)\sqrt{s}}}{s}, s, t\right) \quad \textbf{(20)}$$

```
> u:=subs(infinity=N,Ut):
> u:=subs(N=50,u);
```

$$u := \sum_{n=1}^{50} (-1)^{1+n}\, invlaplace\left(\frac{e^{\sqrt{s}\,(x-2n+1)}}{s}, s, t\right) + \sum_{n=1}^{50} (-1)^{1} \quad \textbf{(21)}$$

$$+ n\, invlaplace\left(\frac{e^{(-x-2n+1)\sqrt{s}}}{s}, s, t\right)$$

```
> plot3d(u,x=1e-6..1,t=1e-6..0.2,axes=boxed,labels=[x,t,"u"],
  orientation=[-150,60]);
```




```
> plot([subs(t=1e-6,u),subs(t=1e-2,u),subs(t=0.05,u),subs(t=0.2,u)
  ],x=0..1,axes=boxed,thickness=5,labels=[x,"u"]);
```


1
0.8
0.6
u
0.4
0.2
0
0
0.2
0.4
0.6
0.8
1
x